Geometry Dependent Behavioral RF Model of Spiral Inductors

Elissaveta Gadjeva[1] and George Valkov[2]

Abstract – **Accurate modeling and simulation of spiral inductors are very important in RFIC design. To enable a more accurate description of the device behavior, the current flowing into the substrate is to be considered. In the present paper a computer wideband behavioral model is developed that considers the substrate losses and is geometry dependent. The model is developed using VHDL-AMS language and accepts the geometry parameters. The model is verified using measurement data in the frequency range from 1GHz to 10GHz and gives a relative error less than 5%. The VHDL-AMS version of the model is simulated in the mixed-language, mixed-signal simulation environment, provided by the Dolphin Integration SMASH simulator.**

Keywords – **Spiral inductor, Wideband behavioral model, Geometry dependent model, VHDL-AMS.**

I. Introduction

A general approach to design and simulation of electronic circuits and systems is to separate the large projects into a hierarchy of smaller modules and sub-modules. On one hand, this allows solving a complex task to be performed by solving many simple tasks, as every simple block can be designed and verified separately from the entire project. On the other hand, many of the modules in the project design can reuse sub-modules, which are already optimized and have a predetermined behavior. Some models can be geometry dependent and parameterized. In this way a given model can be instantiated with different parameters and can represent many elements from the same type, but with different geometric sizes.

Simulators based on standard behavioral languages like VHDL-AMS [1] allow the simulation of models, described at different levels of complexity. Hence a large system design can be described and simulated in low-details (system-level), to verify the functionality of the design, before spending resources to create more detailed sub-modules. Once the system-level performance is verified, additional details can be added, and for each module it is possible to switch between architectures with various levels of complexity. An important feature of the behavior languages is that models can be organized as a library and then reused in many projects, sub-projects and testing environments. Silicon factories can provide models for the most commonly used elements for the available technology processes.

With the development of the instruments for Analog Behavioral Modeling (ABM) and the simulation tools, simulators like Dolphin Integration SMASH allow mixed-signal, multi-level, multi-domain, mixed-language simulations [2]. Hence it is possible to design the various parts of the project in different languages like: *PSpice* language [3], VHDL-AMS, Verilog-AMS [1] and Behavioral-C [4], and to simulate them together.

The physical model of planar spiral inductor on silicon (1-Π) [5] is a very popular model used in microelectronic design. Its model parameter values can be expressed directly using the geometry of the spiral inductor. The skin effect at high frequencies is modeled using a frequency dependent series resistance. This frequency dependent element makes the 1-Π model incompatible with transient analysis and broad-band design. The 2-Π model taking into account the decreasing in the series resistance at higher frequencies is proposed in [6]. A wide-band spiral inductor model is proposed in [7]. Its specific feature is that the model parameters cannot be expressed as a function of inductor geometry. The substrate-coupled equivalent circuit model of an inductor based on substrate phenomena is proposed in [8]. The advantage of this model compared to the 1-Π and 2-Π models, is the prediction of the drop-down characteristic in the series resistance at higher frequencies. An enhancement of the model [8] is the scalable substrate-coupled (*SSC*) model [9] where the parameters of the model elements are expressed as monomial expressions in terms of physical geometry. As a result, the model in [9] combines the advantages of the 1-Π model and the wide-band model [7]: geometry dependence of the model parameters and high accuracy in the whole frequency range.

In the present paper, a computer model coresponding to the geometry dependent, scallable substrate-coupled spiral inductor model [9] is proposed, using the VHDL-AMS language.

II. Model Description

The equivalent electrical circuit of a simplified substrate-coupled spiral inductor is shown in Fig. 1 [9]. The model parameters are geometry dependent. The independent geometry parameters are the following: outer diameter, trace width, spacing between the traces, as well as the number of turns. The model parameters are: capacitance between terminals C_S, capacitances between the wiring and the substrate C_{OX1}, C_{OX2}, substrate capacitances C_{SI1}, C_{SI2}, wiring resistance R_S, silicon resistance R_{SI1}, R_{SI2}, substrate resistance R_{SUB}, wiring inductance L_S, L_{SUB} – representing the eddy current flowing in the substrate, coupling coefficient k of L_S and L_{SUB}, corresponding to the mutual inductance M_S.

[1] Technical University of Sofia, 8, Kliment Ohridski Blvd., 1000 Sofia, BULGARIA, e-mail: egadjeva@tu-sofia.bg
[2] Technical University of Sofia, 8, Kliment Ohridski Blvd., 1000 Sofia, BULGARIA, e-mail: gvalkov@abv.bg



Fig. 1. Simplified substrate-coupled spiral inductor model

The equations describing the geometry dependent model parameters are in the form:

$$C_S = K_1 \cdot n \cdot w^2 \cdot 10^{-15} \quad (1)$$

$$R_S = \frac{K_2 \cdot l}{w} \quad (2)$$

$$C_{OX1} = C_{OX2} = K_3 \cdot l \cdot w \cdot 10^{-15} \quad (3)$$

$$R_{SI1} = R_{SI2} = \frac{K_4}{l \cdot w} \quad (4)$$

$$C_{SI1} = C_{SI2} = K_5 \cdot l \cdot w \cdot 10^{-15} \quad (5)$$

$$L_S = \beta_1 \cdot d_{out}^{a_1} \cdot w^{b_1} \cdot s^{c_1} \cdot n^{d_1} \cdot d_{avg}^{e_1} \cdot 10^{-9} \quad (6)$$

$$L_{SUB} = \beta_2 \cdot d_{out}^{a_2} \cdot w^{b_2} \cdot s^{c_2} \cdot n^{d_2} \cdot \left(L_S \cdot 10^9\right)^{e_2} \cdot 10^{-9} \quad (7)$$

$$R_{SUB} = \beta_3 \cdot n^{a_3} \cdot (w+s)^{b_3} \cdot l^{c_3} \quad (8)$$

$$k = 1 - \exp\left(\beta_4 \cdot n^{a_4} \cdot d_{out}^{b_4} \cdot w^{c_4} \cdot s^{d_4}\right) \quad (9)$$

$$M_S = k \cdot \sqrt{L_S \cdot L_{SUB}} \quad (10)$$

The current that flows in the substrate network composing of L_{SUB} and R_{SUB}, is in the opposite direction to the current that flows in the wiring network composing of L_S and R_S, as a result of mutual inductance M_S [9]. This effect reduces the Equivalent Terminal Resistance (ETR) of the inductor at high frequency and is an improvement over the 1-Π model, for which ETR is not restricted and may reach exceedingly high values. At low frequency, the eddy current has no impact over ETR, due to the low linkage between the wiring and substrate networks, via the high impedance of C_{OX1} and C_{OX2}.

The equations for intermediate parameters – inner diameter d_{in}, average diameter d_{avg} and conductor length l, are in the form:

$$d_{in} = d_{out} \cdot \left(n \cdot (s+w) - s\right) \quad (11)$$

$$d_{avg} = \frac{d_{in} + d_{out}}{2} \quad (12)$$

$$l = 4 \cdot d_{avg} \cdot n \quad (13)$$

The independent model parameters are: outer diameter of the coil d_{out}, conductor width w, space between the traces s, number of turns n. Additional process dependent constants required to calculate the model parameters are shown in Table I and Table II [9].

TABLE I
PROCESS DEPENDENT CONSTANTS FOR CMOS 0.35μm

i	**1**	**2**	**3**	**4**	**5**
β_i	2.50E-4	6.18E-7	156	-4.85E+4	
a_i	1.84	0.94	1.36	0.91	
b_i	-0.76	4.13	0.93	-1.96	
c_i	-0.14	-1.06	-1.40	-0.83	
d_i	1.10	-1.90	0	0.56	
e_i	0	1.35	0	0	
K_i	0.0415	0.0302	4.28E-3	8.04E+6	2.10E-4

TABLE II
PROCESS DEPENDENT CONSTANTS FOR SOI 0.15μm

i	**1**	**2**	**3**	**4**	**5**
β_i	1.00E-4	1.00E-6	39	-4.90E+4	
a_i	2.31	0.98	1.30	2.60	
b_i	-1.47	4.39	0.93	-2.12	
c_i	-0.08	-0.99	-1.40	-0.98	
d_i	1.24	-1.84	0	1.60	
e_i	0	0.68	0	0	
K_i	0.0895	0.0466	2.93E-3	3.03E+7	2.10E-4

III. MODEL IMPLEMENTATION IN VHDL-AMS

The VHDL-AMS description of the model has the following structure:

1. Define a list of libraries used by the model (electrical systems and mathematical functions). VHDL-AMS description is in the form:

```
library IEEE;
  use IEEE.electrical_systems.all;
  use IEEE.math_real.all;
entity inductor_pi is
```

2. Define a list of independent geometry parameters of the model: outer diameter of the coil d_{out}, conductor width w, space between the traces s, number of turns n.

```
  generic (
    dout: real := 250.0;
    w   : real :=  10.0;
    s   : real :=   5.0;
    n   : real :=   5.0
  );
```

3. Define a list of electrical terminals. The model interfaces to the external world via two general electrical terminals *n1* and *n2*. In addition, there is an internal connection to the node called *electrical_ref* (ground). It has global visibility and does not need to be explicitly declared in the port list.

```
  port (terminal n1, n2 : electrical);
end entity inductor_pi;
architecture ideal of inductor_pi is
```

4. In order to reduce the number of declarations and to make the code more readable, the process dependent constants are declared as arrays. The implementation code that follows, declares the constants for CMOS 0.35µm, Table I [1]. The constants for SOI 0.15µm fabrication process are available in Table II.

```
constant beta: real_vector(1 to 4) := ( 2.50e-4,
 6.18e-7, 156.0,  -4.85e+4   );
constant aa  : real_vector(1 to 4) := ( 1.84,
 0.94,  1.36,  0.91 );
constant bb  : real_vector(1 to 4) := (-0.76,
 4.13,  0.93, -1.96 );
constant cc  : real_vector(1 to 4) := (-0.14,
-1.06,  1.40, -0.83 );
constant dd  : real_vector(1 to 4) := ( 1.10,
-1.90,  0.00,  0.56 );
constant ee  : real_vector(1 to 4) := ( 0.00,
 1.35,  0.00,  0.00 );
constant KK  : real_vector(1 to 5) := ( 0.0415,
 0.0302,  4.28e-3, 8.04e+6, 2.10e-4);
```

5. Define and calculate the equations (11)-(13) for intermediate parameters: inner diameter d_{in}, average diameter d_{avg} and conductor length l.

```
constant din :real:=dout-2.0*(n*(s+w)-s);
constant davg:real:= 0.5*( din + dout );
constant l   :real:= 4.0*davg*n;
```

6. Define and calculate the equations (1)-(10) for schematic parameters: C_S, C_{OX1}, C_{OX2}, C_{SI1}, C_{SI2}, R_S, R_{SI1}, R_{SI2}, R_{SUB}, L_S, L_{SUB}, k and M_S.

```
constant Cs  : capacitance := KK(1)* n*w*w *1.0e-15;
constant Rs  : resistance  := KK(2)* l/w;
constant Cox1: capacitance := KK(3)* l*w   *1.0e-15;
constant Cox2: capacitance := Cox1;
constant Rsi1: resistance  := KK(4)/(l*w);
constant Rsi2: resistance  := Rsi1;
constant Gsi1: real        := 1.0/Rsi1;
constant Gsi2: real        := 1.0/Rsi2;
constant Csi1: capacitance := KK(5)*l*w    *1.0e-15;
constant Csi2: capacitance := Csi1;
constant Ls  : inductance  := beta(1)*(dout**aa(1))*
 (w**bb(1))*(s**cc(1))*(n**dd(1))*(davg**ee(1))
 *1.0e-9;
constant Lsub: inductance  := beta(2) *(dout**aa(2))
 *(w**bb(2))*(s**cc(2))*(n**dd(2))
 *((Ls*1.0e+9)**ee(2))*1.0e-9;
constant Rsub: resistance  := beta(3)*(n**aa(3))
 *((w+s)**bb(3))*(l**cc(3));
constant k   : real        := 1.0-exp(beta(4)
 *(n**aa(4))*(dout **bb(4))*(w**cc(4))*(s**dd(4)));
constant Ms  : inductance  := k*sqrt(Ls*Lsub);
```

7. Define a list of internal and external electrical nodes: n_ox1 between C_{OX1} and C_{SI1}, n_ox2 between C_{OX2} and C_{SI2}, n_s between L_S and R_S.

```
terminal n_ox1, n_ox2, n_s : electrical;
```

8. Define a list of quantities to describe the model equations: voltage U across the external nodes $n1$ and $n2$, current I_{CS} flowing through the capacitor C_S, current I_{LS} flowing through the inductor L_S and the resistor R_S, voltage U_{SUB} across L_{SUB} and R_{SUB}, current I_{SUB} through L_{SUB} and R_{SUB}, voltages U_{OX1} and U_{OX2} across capacitors C_{OX1} and C_{OX2} respectively, currents I_{OX1} and I_{OX2} through capacitors C_{OX1} and C_{OX2} respectively, voltages U_{SI1} and U_{SI2} across capacitors C_{SI1} and C_{SI2} respectively, current I_{SI1} through capacitor C_{SI1} and resistor R_{SI1}, current I_{SI2} through capacitor C_{SI2} and resistor R_{SI2}.

```
quantity U    across Ics  through n2 to n1;
quantity             Ils  through n2 to n1;
quantity Usub across Isub through
                 n_ox2 to n_ox1;
quantity Uox1 across Iox1 through n1 to n_ox1;
quantity Uox2 across Iox2 through n2 to n_ox2;
quantity Usi1 across Isi1 through
                 n_ox1 to electrical_ref;
quantity Usi2 across Isi2 through
                 n_ox2 to electrical_ref;
```

9. Define the set of equations to represent the schematic in Fig. 1. Some of the equations represent more than one schematic component, e.g. for the two parallel branches between *n_ox1* and *electrical_ref* (ground), the current I_{OX1} is a sum of two currents: one flowing through the resistor R_{SI1} and the other flowing through the capacitor C_{SI1}. This is used for better readability of the code and to reduce the number of equations.

```
begin
    Ics  == Cs   *U   'dot;
    Iox1 == Cox1*Uox1'dot;
    Iox2 == Cox2*Uox2'dot;
    Isi1 == Csi1*Usi1'dot + Usi1*Gsi1;
    Isi2 == Csi2*Usi2'dot + Usi2*Gsi2;
    U    == Ls   *Ils 'dot
          + Ms   *Isub'dot + Ils *Rs;
    Usub == Lsub*Isub'dot
          + Ms   *Ils 'dot + Isub*Rsub;
end architecture;
```

IV. Computational Results of an Example Inductor

To verify the VHDL-AMS implementation of the model, the simulation results for the quality factor have been compared to the measurement results for an example inductor fabricated on the CMOS 0.35µm process [9]. The dimensions of the inductor are shown in Table III. The simulation and measurement results for Q are shown in Fig. 2. The relative error does not exceed 5%.

TABLE III
DIMENSIONS OF INDUCTOR FABRICATED ON CMOS 0.35µm PROCESS

d_{out} [µm]	w [µm]	s [µm]	n
250	10	5	5



Fig. 2. Simulation and measurement results for Q



Fig. 3. Simulation results for *Zeq*



Fig. 4. Simulation results for ETR



Fig. 5. Simulation results for L_{eq} for 1-Π, 2-Π, *SC* and *SSC* models



Fig. 6. Simulation results of *Q* for 1-Π, 2-Π, *SC* and *SSC* models and comparison to the measurement results

The simulation results for the module Z_{eq} of the equivalent impedance of the spiral inductor, for ETR and L_{eq} are shown in Fig. 3, Fig. 4 and Fig. 5 respectively, for 1-Π, 2-Π and scalable substrate-coupled (*SSC*) models. The simulation results of *Q* for 1-Π, 2-Π and *SSC* models are compared to the measurement data (Fig. 6). The advantages of the scalable substrate-coupled model to the other models are seen: a high accuracy in the whole frequency range, as well as its scalability.

The proposed computer realization of the behavioral scalable substrate-coupled model of the spiral inductor using VHDL-AMS language is portable and can be used in various simulation environments.

V. Conclusion

A VHDL-AMS behavioral substrate-coupled model of spiral inductor has been developed in the paper. The model considers substrate losses and is valid in a wide frequency range. It is characterized by geometry dependence of the model parameters. The proposed computer model is verified to measurement data in the frequency range of 1GHz to 10GHz and gives a relative error less than 5%. The VHDL-AMS model is simulated in the mixed-language, mixed-signal simulation environment, provided by the Dolphin Integration SMASH simulator.

Acknowledgement

The investigations are supported by the project DTK02/50/17.12.2009.

References

[1] F. Pêcheux, C. Lallement, A. Vachoux, "VHDL-AMS and Verilog-AMS as Alternative Hardware Description Languages for Efficient Modeling of Multi-Discipline Systems", Paper #1446, IEEE, ISSN 0278-0070, 2005.
[2] http://www.dolphin.fr/medal/smash/smash_overview.php
[3] *PSpice A/D Reference Guide*, Cadence Design Systems, 2009.
[4] M. Chung, S. Na, C. Kyung, "System-Level Performance Analysis of Embedded System Using Behavioral C/C++ Model", 0-7803-9060-1/05/$20.00, IEEE, pp. 178-191, 2005.
[5] C. P. Yue, C. Ryu, J. Lau, T. H. Lee and S. S. Wong, "A physical model for planar spiral inductors on silicon", Proc. IEEE Int. Electron Devices Meeting Tech. Dig., San Francisco, pp. 155-158, CA, Dec. 1996.
[6] Y. Cao, R. A. Groves, X. Huang, N. Zamdmer, J. Plouchart, R. Wachnik, T.-J. King and C. Hu, "Frequency-independent equivalent-circuit model for on-chip spiral inductors", IEEE J. Solid-State Circuits, vol. 38, no. 3, pp. 419–426, Mar. 2003.
[7] J. Gil, H. Shin, "A simple wide-band on-chip inductor model for silicon-based RF Ics", IEEE Transactions on Microwave Theory and Techniques, vol. 51, n. 9, pp. 2023–2028, Sept., 2003.
[8] Minoru Fujishima, Jun Kino, "Accurate Subcircuit Model of an On-Chip Inductor with a New Substrate Network", IEEE Symposium on VLSI Circuits, pp. 376-379, 2004.
[9] Ivan C. H. Lai, Minoru Fujishima, "A new on-chip substrate-coupled inductor model, implemented with scalable expressions", IEEE journal of solid-state circuits, vol. 41, pp. 2491-2499, 11 November 2006.